Eulerin yhtälön muuntaminen vakiokertoimiseksi

*Maple*a voidaan käyttää yleisen Eulerin yhtälön muuntamiseen vakiokertoimiseksi yhtälöksi sijoituksella $x = \mathbf{e}^t$. Seuraava esitys toimii periaatteessa mille tahansa kertaluvulle n. Jos $n > 20$, alkaa laskenta kuitenkin vaatia varsin paljon resursseja.

Syötetään Eulerin yhtälön kertaluku ja muodostetaan yhtälö:

```
> n:= 5;
```

$$n := 5$$

```
> euleryht:= sum(a[k]*x^k*(D@@k)(y)(x), k=0..n)=0;
```

$$euleryht := a_0\, \mathrm{y}(x) + a_1\, x\, \mathrm{D}(y)(x) + a_2\, x^2\, (\mathrm{D}^{(2)})(y)(x) + a_3\, x^3\, (\mathrm{D}^{(3)})(y)(x) + a_4\, x^4\, (\mathrm{D}^{(4)})(y)(x) + a_5\, x^5\, (\mathrm{D}^{(5)})(y)(x) = 0$$

Sijoituksen seurauksena syntyy uusi tuntematon funktio $\mathrm{u}(t) = \mathrm{y}(\mathbf{e}^t)$. Funktiot y ja u toisiinsa sitova yhtälö talletetaan nimelle yht_0:

```
> yht[0]:= u(t)=y(exp(t));
```

$$yht_0 := \mathrm{u}(t) = \mathrm{y}(\mathbf{e}^t)$$

Tätä yhtälöä derivoidaan muuttujan t suhteen n kertaa, jotta saadaan vastaavat derivaattojen väliset yhtälöt; nämä kerätään listaksi:

```
> yhtalot:= {seq(diff(yht[0], [t$k]), k=0..n)};
```

$$\begin{aligned} yhtalot := \{ & \frac{d}{dt}\mathrm{u}(t) = \mathrm{D}(y)(\mathbf{e}^t)\, \mathbf{e}^t,\ \mathrm{u}(t) = \mathrm{y}(\mathbf{e}^t),\ \frac{d^5}{dt^5}\mathrm{u}(t) = (\mathrm{D}^{(5)})(y)(\mathbf{e}^t)\,(\mathbf{e}^t)^5 \\ & + 10\,(\mathrm{D}^{(4)})(y)(\mathbf{e}^t)\,(\mathbf{e}^t)^4 + 25\,(\mathrm{D}^{(3)})(y)(\mathbf{e}^t)\,(\mathbf{e}^t)^3 + 15\,(\mathrm{D}^{(2)})(y)(\mathbf{e}^t)\,(\mathbf{e}^t)^2 + \mathrm{D}(y)(\mathbf{e}^t)\,\mathbf{e}^t, \\ & \frac{d^4}{dt^4}\mathrm{u}(t) = (\mathrm{D}^{(4)})(y)(\mathbf{e}^t)\,(\mathbf{e}^t)^4 + 6\,(\mathrm{D}^{(3)})(y)(\mathbf{e}^t)\,(\mathbf{e}^t)^3 + 7\,(\mathrm{D}^{(2)})(y)(\mathbf{e}^t)\,(\mathbf{e}^t)^2 + \mathrm{D}(y)(\mathbf{e}^t)\,\mathbf{e}^t, \\ & \frac{d^3}{dt^3}\mathrm{u}(t) = (\mathrm{D}^{(3)})(y)(\mathbf{e}^t)\,(\mathbf{e}^t)^3 + 3\,(\mathrm{D}^{(2)})(y)(\mathbf{e}^t)\,(\mathbf{e}^t)^2 + \mathrm{D}(y)(\mathbf{e}^t)\,\mathbf{e}^t, \\ & \frac{d^2}{dt^2}\mathrm{u}(t) = (\mathrm{D}^{(2)})(y)(\mathbf{e}^t)\,(\mathbf{e}^t)^2 + \mathrm{D}(y)(\mathbf{e}^t)\,\mathbf{e}^t \} \end{aligned}$$

Tuntemattomat, ts. funktion y derivaatat kerätään omaksi listakseen; muuttujaksi otetaan t:

```
> tuntemattomat:= {seq((D@@k)(y)(exp(t)), k=0..n)};
```

$$tuntemattomat := \{\mathrm{y}(\mathbf{e}^t),\ \mathrm{D}(y)(\mathbf{e}^t),\ (\mathrm{D}^{(2)})(y)(\mathbf{e}^t),\ (\mathrm{D}^{(3)})(y)(\mathbf{e}^t),\ (\mathrm{D}^{(4)})(y)(\mathbf{e}^t),\ (\mathrm{D}^{(5)})(y)(\mathbf{e}^t)\}$$

Jotta alkuperäinen differentiaaliyhtälö saadaan muunnetuksi funktiota u koskevaksi, yhtälöryhmästä ratkaistaan funktion y derivaatat ja sijoitetaan nämä differentiaaliyhtälöön; muuttujaksi otetaan t:

```
> sij:= solve(yhtalot, tuntemattomat);
```

$$sij := \left\{ (D^{(5)})(y)(e^t) = \frac{\left(\frac{d^5}{dt^5}u(t)\right) - 10\left(\frac{d^4}{dt^4}u(t)\right) + 35\left(\frac{d^3}{dt^3}u(t)\right) - 50\left(\frac{d^2}{dt^2}u(t)\right) + 24\left(\frac{d}{dt}u(t)\right)}{(e^t)^5}, \right.$$
$$(D^{(4)})(y)(e^t) = -\frac{-\left(\frac{d^4}{dt^4}u(t)\right) + 6\left(\frac{d^3}{dt^3}u(t)\right) - 11\left(\frac{d^2}{dt^2}u(t)\right) + 6\left(\frac{d}{dt}u(t)\right)}{(e^t)^4},$$
$$(D^{(2)})(y)(e^t) = -\frac{-\left(\frac{d^2}{dt^2}u(t)\right) + \left(\frac{d}{dt}u(t)\right)}{(e^t)^2},$$
$$\left. (D^{(3)})(y)(e^t) = \frac{\left(\frac{d^3}{dt^3}u(t)\right) - 3\left(\frac{d^2}{dt^2}u(t)\right) + 2\left(\frac{d}{dt}u(t)\right)}{(e^t)^3}, y(e^t) = u(t), D(y)(e^t) = \frac{\frac{d}{dt}u(t)}{e^t} \right\}$$

```
> subs(x=exp(t), euleryht):
  vakiokertyht:= subs(sij, %):
  simplify(%);
```

$$a_0 u(t) + a_1\left(\frac{d}{dt}u(t)\right) + a_2\left(\frac{d^2}{dt^2}u(t)\right) - a_2\left(\frac{d}{dt}u(t)\right) + a_3\left(\frac{d^3}{dt^3}u(t)\right) - 3a_3\left(\frac{d^2}{dt^2}u(t)\right)$$
$$+ 2a_3\left(\frac{d}{dt}u(t)\right) + a_4\left(\frac{d^4}{dt^4}u(t)\right) - 6a_4\left(\frac{d^3}{dt^3}u(t)\right) + 11a_4\left(\frac{d^2}{dt^2}u(t)\right) - 6a_4\left(\frac{d}{dt}u(t)\right)$$
$$+ a_5\left(\frac{d^5}{dt^5}u(t)\right) - 10a_5\left(\frac{d^4}{dt^4}u(t)\right) + 35a_5\left(\frac{d^3}{dt^3}u(t)\right) - 50a_5\left(\frac{d^2}{dt^2}u(t)\right) + 24a_5\left(\frac{d}{dt}u(t)\right) = 0$$

Tällöin on saatu Eulerin yhtälöä vastaava vakiokertoiminen yhtälö. Jotta tämä hahmottuisi selkeämmin, termit kootaan funktion u derivaattojen mukaan ryhmiteltyinä:

```
> derivaatat:=[seq(diff(u(t), [t$k]), k=0..n)];
```

$$derivaatat := \left[u(t), \frac{d}{dt}u(t), \frac{d^2}{dt^2}u(t), \frac{d^3}{dt^3}u(t), \frac{d^4}{dt^4}u(t), \frac{d^5}{dt^5}u(t)\right]$$

```
> ryhmitettyvakiokertyht:= collect(vakiokertyht, derivaatat);
```

$$ryhmitettyvakiokertyht := a_0\,\mathrm{u}(t) + (a_1 - a_2 - 6\,a_4 + 2\,a_3 + 24\,a_5)\left(\frac{d}{dt}\,\mathrm{u}(t)\right) + (-3\,a_3 + a_2 + 11\,a_4 - 50\,a_5)\left(\frac{d^2}{dt^2}\,\mathrm{u}(t)\right) + (-6\,a_4 + 35\,a_5 + a_3)\left(\frac{d^3}{dt^3}\,\mathrm{u}(t)\right) + (a_4 - 10\,a_5)\left(\frac{d^4}{dt^4}\,\mathrm{u}(t)\right) + a_5\left(\frac{d^5}{dt^5}\,\mathrm{u}(t)\right) = 0$$

Saadun yhtälön vasemmasta puolesta voidaan myös poimia eri derivaattojen kertoimet:

```
> coeffs(lhs(ryhmitettyvakiokertyht), derivaatat);
```

$$a_0,\ a_1 - a_2 - 6\,a_4 + 2\,a_3 + 24\,a_5,\ -3\,a_3 + a_2 + 11\,a_4 - 50\,a_5,\ -6\,a_4 + 35\,a_5 + a_3,\ a_4 - 10\,a_5,\ a_5$$

```
>
```

Kertoimet ovat todellakin vakioita.

Linkit

Eulerin yhtälön muuntaminen vakiokertoimiseksi ja ratkaiseminen
yhtälön muuntaminen sijoituksella

SKK & MS 31.05.2001